


Internet Scripting ! (QuarkXPress Scripting 入門番外編 !)

「テレホーダイ WWW サーバー」を作ろう

鎌田幸雄

OCNの登場で従来よりはるかに安く専用線でのインターネット接続が可能になってきました。マック関連の雑誌でも、個人のマックでサーバーを運用する記事を目にするようになってきました。しかし、数万円まで安くなったとはいえ個人で負担することはなかなかできません (何といても年間の通信料だけで、最新マックが購

入できる金額ですからね)。「サーバーなんて専用線の話、ダイヤルアップ接続している自分には関係ないさ」と考えていませんか? ホームページを持てるプロバイダーと契約していれば、簡単に、しかも、月々数千円で可能です。今月は、QuarkXPress Scripting 入門をお休みして、Internet Scripting に挑戦してみましょう!



ダイヤルアップ接続したマックをサーバーにできるの?

専用線につながってなくても、ダイヤルアップで接続すれば、IPアドレスが割り当てられます。WWWサーバーを起動したならば、アクセスする人は、このIPアドレスを入力することで、自分のサーバーに接続できます。理屈は簡単なのです。しかし、ダイヤルアップ接続では、IPアドレスが接続するたびに変わってしまいます。ですから、アクセスしてくれる人に、毎回変わってしまうIPアドレスをどのように伝えるかが問題になります。

ホームページを開設できれば、そのHTMLに自分のIPアドレスを付けたリンク先を書き込んでおき、自分のマックでサーバーを立ち上げることで実現できます。NTTのテレホーダイと契約すれば、夜11時から翌朝8時までは、つなぎっぱなしにしても定額で済みます。テレホーダイを利用して、時間限定サーバーを作り、毎晩11時にIPアドレスを取得して、HTMLを書き換えて、アップロードしてサーバーを立ち上げる作業をAppleScriptで自動化してしまおうというのが今回のテーマです。

自分のマックをサーバーにするメリットは?

時間は限定されますが、何よりも安価でサーバーを持って、いろいろな実験ができることです。CGIもAppleScriptが利用できますし、MacPerlでも手軽に作ることができます。一般的な商用サーバーはUNIXがほとんどですから、Macintoshで行うことは意義があるでしょう。本格的に専用線を引く前段階として、ノウハウを蓄積するには最適だと考えています。

事前にやっておくことは?

- (1)ホームページを開設できるインターネット接続プロバイダーと契約をしましょう。従量制ではなくて、定額制にします。
- (2)NTTのテレホーダイに契約しましょう。
- (3)ホームページを作っておきましょう。その中に、ここをクリックすると、私のホストへ接続できますよ! のリンクを埋め込む場所を用意しておきます。

ログオンからログオフまでの手順

- 1) 毎晩11時に起動する
- 2) PPPで自動接続する
- 3) IPアドレスを取得する
- 4) ホームページのHTMLのテキストを3)で取得したアドレスに書き換える
- 5) 書き換えたHTMLをFetchでアップロードする
- 6) サーバーを起動する
- 7) 翌朝8時に終了する (HTMLに終了メッセージを書き込み、アップロードしてから終了する)。

1)は時間がくれば自動起動するようなスクリプトを組むか、市販あるいはフリーウェアに、自動起動するものがありますので、それらを利用すれば問題ありません。7)も同様にできるでしょう。

4)は、AppleScriptには、標準でスクリプタブルなエディタが付いてきます。「スクリプティング対応エディタ」がそうですので、それを利用します。

5)は、Fetchがバージョン3.0でAppleScript対応したので問題ありません。

6)は、スクリプト的なWWWサーバーソフトがあります。となると、課題は2)と3)です。

PPPの自動接続



スクリプト的なPPPがあるのが問題になります。昨年、OpenTransport対応かつスクリプト的なPPPがAppleから提供されました。OpenTransport/PPP1.0です（動作条件がOpenTransport1.1.1以上です。サンプルスクリプトも収録されています）。これを利用すれば自動接続は簡単にできます。

IPアドレスの取得

Open Transport/PPPで接続すると、IPアドレスが表示されるのですが、これは接続先のアドレスであって、自分に割り当てられたアドレスではありません。それではどうしましょう...

これについては、2つの解決方法がありました。

(1) additionを利用する

シェアウェアで、TCP/IP Scripting Additionがあります。get tcp my address とするとIPアドレスが取得可能です。



(2)フリーウェアを利用する

NetEvents というのがあります。これが、TCP/IP Scripting Additionと同様のことができます。v1.06が最新でこれを利用しますが、IPアドレスの取得以外にもいろいろとスクリプトできるのですが、このサンプルスクリプトがv1.06では同梱されていません。これを取得したい方は前バージョンのv1.04を取得してください。

スクリプト例

PPP自動接続のスクリプト

```
set username to "xxxxxxx"
set password to "xxxxxxx"
set telephone to "0423-12-3456"
-----
repeat until ((state of (PPP status)) is "connected")
  try
    PPP connect address telephone user name (username)
  password (!password)
  my timeCount(5)
  on error
  end try
end repeat
-----
on timeCount(s)
  set intervalSecond to s
  set date0 to current date
  repeat while ((current date) - date0) < intervalSecond
  end repeat
end timeCount
```

変数 *username*、*password* は、接続プロバイダーへのアカウントとパスワードです。変数 *telephone* は、接続ポイントの電話番号です。*timeCount* サブルーチンは、指定秒間ループします。

IPアドレス取得のスクリプト

(1) NetEvents を使用した場合

```
tell application "NetEvents 1.0b6"
  set NEW_AD_TXT to AddrToName (MyIPAddress)
end tell
```

(2) TCP/IP Scripting Addition を使用した場合

```
set NEW_AD_TXT to tcp address to name (tcp my address)
```

(1)、(2)とも、変数 *NEW_AD_TXT* に IP アドレスの名称が入ります。

HTML ファイルの書き換え

```
tell application "スクリプティング対応エディタ"
  open alias "VegaHD:/x/sunrise:/index.html"
  tell document 1
    tell paragraph 77
      cut
      make paragraph with data "red_arrow.gif"
    end tell
    tell paragraph 80
      cut
      make paragraph with data NEW_AD_TXT
    end tell
    tell paragraph 83
      cut
      make paragraph with data "現在アクセス可能¥, 翌朝
07:50まで!"
    end tell
    save in file "VegaHD:/x/sunrise:/index.html" as text
    close saving no
  end tell
end tell
```

 Midnight Server (23:10~07:50)
(現在アクセス可能, 翌朝07:50まで!)

ハードディスク "VegaHD" の中のフォルダ "/x/sunrise/" の中に書き換える html 書類, "index.html" があります。これの 77 行目を "red_arrow.gif", 80 行目を先に取得した変数 *NEW_AD_TXT*, 83 行目を, "現在アクセス可能, 翌朝 07:50 まで" に書き換えるわけです。ちなみに, index.html の 80 行目の前後は下記のようになっています。

```
<A HREF="http://
p3ngym068.highway.or.jp
">Midnight Server </A><FONT SIZE=-1>(23:10 ~
07:50)</FONT><BR>
```

80行目

イタリックの部分が IP アドレスです。ここが毎回書き換えられます。これで、ホームページの "Midnight Server" をクリックすると、自分のマックにリンクを張ることができました。

Fetch によるアップロード

Fetch の環境設定によって、多少変わってきます。Fetch を起動したときに、ショートカットウィンドウが開く設定になっているか否かで、どちらかを使い分けることになります。

(1) ショートカットウィンドウが表示されない場合

```
set DestUrl to "ftp://99999999.*****@www.age.or.jp/"
set doc to alias "VegaHD:/x/sunrise:/index.html"
tell application "Fetch 3.0.1"
  activate
  put into url DestUrl item doc
  quit
end tell
```

99999999 は ftp に入るためのアカウント, ***** はそのパス

ワードです。www.age.or.jpはプロバイダーのアドレスになります。
(2)ショートカットウィンドウが表示される場合

```
tell application "Fetch 3.0.1"
  tell shortcut window 1
    open bookmark "NetAge"
  end tell
  tell transfer window 1
    put into item alias "VegaHD:/x/sunrise:/index.html"
  end tell
end tell
```

これは、ショートカットウィンドウに名称、アカウント、パスワードが入っていることが前提です。

サーバーソフトウェア

あとはサーバーを立ちあげるだけです。では、どんなものがよいのでしょうか？ 試してきたものを、独断的な感想を述べたいと思います。

- (1) Personal Web Sharing
今後、標準搭載されるパーソナルサーバ。機能拡張とコントロールパネルにいれる。
- (2) Personal Web Server
これも、機能拡張とコントロールパネルにいれるもの。しかし、簡単にホームページができる配慮がなされている。手軽さでは一番。
しかしながら、PersonalWeb Sharingとともに、アプリケーションではないので、AppleScriptで制御ができません。
- (3) NetPresents4.1
アプリケーションだが、設定が独特。AppleScript未対応。
- (4) Pictorius Net Servers
アプリケーションだが、シェアウェア。設定もやや難しい。しかしそのぶん高機能。AppleScript未対応。
- (5) MacHTTP2.2.2
アプリケーションで、しかもフリーウェア。操作も簡単！AppleScriptにも対応しているし、いろんな資料もそろっている(つまり多くの人たちが使用している)ので最適。
- (6) WebStar 2.0
これは、試していません。製品で、高価だからです。Personal WebStarを期待していたのですが、残念ながら開発中止したようです。
- (7) Quid Pro Quo 1.0
アプリケーションで、しかもフリーウェア。AppleScriptにも対応している。基本的にMacHTTPの後発ライバル製品という感じ。操作も簡単です。ただ、起動するだけです。後発の強みで、MacHTTPより操作性は良くなっていると思います。
テレホーダイWWWサーバーはQuid Pro Quoで運用していきたいと思います。

サーバーを起動するスクリプトは、

```
tell application "Quid Pro Quo" to activate
```


これだけで完了です。Quid Pro Quoの同一階層にdefault.htmlを用意すれば、それが起動時に開かれるhtml書類となります。

実際に運用してみたら・・・

これらのスクリプトをまとめて試験的に運用してみました。い


くつか、問題点が出てきました。
上記のスクリプトは、エラーが何も起こらない時のものです。実際には、ビジーによる再接続や、下記の場合のエラー回避を考慮しないとイケません。

- (1)IPアドレスの取得がうまくできない場合がある！
接続は完了しているのに、IPアドレスの取得でエラーとなる場合があります。この場合、接続からやり直さないといけません。
- (2)HTMLのアップロードに時間がかかってタイムアウト！
Fetchによるアップロードは、回線の混雑状況によっては、時間がかかる場合があります。AppleScriptでは、デフォルトで60秒間応答がないとタイムアウトになりますから、その時間を5分程度に設定し直す必要があるでしょう。
- (3)プロバイダー側から接続解除されてしまう！
筆者の契約しているプロバイダーは約10分間応答がないと、自動的に接続解除するようです。そのための対策として、Webブラウザをスクリプトで操作して、10分ごとに、自分のマックを指定するようにしました。

```
tell application "インターネットエクスプローラ 3.0.1"
  Activate
  repeat
    OpenURL "http://" & NEW_AD_TXT to Window 2 Flags 1
    my timeCount(600)
  end repeat
  quit
end tell
```

終了時のHTML書き換え

```
tell application "スクリプティング対応エディタ"
  open alias "VegaHD:/x/sunrise:/index.html"
  tell document 1
    tell paragraph 77
      cut
      make paragraph with data "black_arrow.gif"
    end tell
    tell paragraph 80
      cut
      make paragraph with data "sorry.html"
    end tell
    tell paragraph 83
      cut
      make paragraph with data "現在アクセス不可, 23:10
      まで待て!"
    end tell
    save in file "VegaHD:/x/sunrise:/index.html" as text
    close saving no
  end tell
end tell
```



ホームページに、アクセス終了のメッセージを表示するよう書き換えます。あとは、起動しているアプリケーションを終了させて完了です。

Netscape NavigatorかInternet Explorerか？

WEBブラウザは、どちらでも結構です。OpenURLのコマンドはどちらもサポートしているので問題はありませぬ。ただし、筆者は次の理由で、Internet Explorerにしています。
・メモリの使用量
Navigator3.01は12000Kを要求します。メモリの少ない筆者の

68Kマックでは負担になります。Internet Explorer3.01は4096Kです。

・ Windowの指定が有効

NavigatorはWindowの指定が、コーディング上はできても実際は有効になりません。常に一番上のウインドウにアクセスします。そのため、ネットサーフィン中に10分経過して自分のマックにアクセスする際、他のホームページへのアクセスが中断されてしまいます。Internet ExplorerのWindowの指定は有効で、最初に設定したウインドウに常にアクセスしていくので、10分間隔でアクセスするウインドウを背後に置いて、もう一つのウインドウを開いてネットサーフィンすることが可能です。

自動起動するためには

"TeleHodai Server"は、時間がくれば自動的に終了するようにしていますので、夜11時に自動起動させればいいのですが、その方法にはいくつかあります。筆者は、シェアウェアのcronを利用しています。on idle ~ end idleのスク립トを組んでの起動も可能ですし、Jterm3.0に付いているJランチ3.0のような自動ランチツールを使用してもよいと思います。

cronを利用した場合

crontabに与えるパラメータ例



```
0 23 * * * nobody "VegaHD:TeleHodai Server"
```

ハードディスク "VegaHD"の第一階層に"TeleHodai Server"がある場合のパラメータです。cronは、他のパラメータや、AppleScriptと組み合わせることによって、非常に有効なツールになります。これについては、機会があれば述べてみたいと思います。

on idle ~ end idleを利用した場合

夜11時を過ぎると、テレホーダイサーバー起動スク립ト "TeleHodai Server"を起動するスク립ト例

```
global strtTime, endTime
set strtTime to 23 * hours + 0 * minutes
set endTime to 7 * hours + 50 * minutes
on idle
  set date1 to time of date ((current date) as text)
  tell application "Finder"
    set ING_APPL to every process
  end tell
  if not ((endTime < date1) and (date1 < strtTime)) then
    if "TeleHodai Server" is not in ING_APPL then
      with timeout of 3000 seconds
        tell application "TeleHodai Server"
          activate
        end tell
      end timeout
    end if
  end if
  return 30
end idle
```

補足 これを起動すると30秒ごとに、on idle ~ end idleの部分が実行されます。現在時刻が夜11時~朝7時50分の間ならば、起動中のアプリケーションをチェックし、"TeleHodai Server"が起動していない場合に実行するものです。保存時には、「実行後、自動的に終了しない」をチェックしないといけません。

TeleHodai Server

次ページのリストのTeleHodai Serverは、まだいろいろ試行錯誤

誤中のスク립トです。夜11時に起動して、朝7時50分に終了します。約2カ月運用してきましたので、実用には耐えうらと思います。筆者の環境はISDNですが、アナログの場合には、待機秒の調整が必要かもしれません。プロバイダーによって巡回時間の変更が必要になるでしょう。その他on idle ~ end idleの中に組み込んだり、cronの他のパラメータと組み合わせて自分なりのスク립トにするのもいいでしょう。また、予期せぬエラーに対応させたりすることが必要になるかもしれません。

アプリケーション入手先リスト

OpenTransport/PPP

OpenTransport

<ftp://ftp.info.apple.com/Apple.Support.Area/>

Apple_SW_Updates/US/Macintosh/

Networking-Communications/Open_Transport/

TCP/IP Scripting Addition

<http://www.mangotree.com/tcpscrippadd.html>

NetEvents

<http://www.biap.com/downloads/netevents.html>

Fetch

<http://www.dartmouth.edu/pages/softdev/fetch.html>

Quid Pro Quo

<http://www.slaphappy.com/>

Netscape Navigator

<http://home.netscape.com/comprod/mirror/>

client_download.html

Internet Explorer

<http://www.microsoft.com/ie/download/>

cron

<http://gargravarr.cc.utexas.edu/cron/cron.html>

最後に

くれぐれも、テレホーダイの適用開始日と電話番号を間違えないように、(^0^)/

CD-ROM収録

・ スクリプト

TeleHodai Server

PPP Connect

PPP Disconnect

PPP Reconnect

・ QuickTimeムービー

TeleHodai Server 起動.mov

・ 筆者のホームページのindex.html

--TeleHodai Server --

```
global date1, strTime, endTime
set username to "xxxxxxx" --(1)
set password to "abcdefgh" --(2)
set telephone to "0423-56-7004"--(3)
set NEW_AD_TXT to ""
set strTime to 23 * hours + 0 * minutes --(4)
set endTime to 7 * hours + 50 * minutes --(5)
```

下記の部分を訂正して下さい
(1)接続ID
(2)接続パスワード
(3)接続電話番号
(4)接続開始時間
(5)接続終了時間
(6)FTP 接続IDとパスワードと接続プロバイダ

```
-----PPP 起動( 回線接続 )
repeat until ((state of (PPP status)) is "connected") and (NEW_AD_TXT is not "")
  try
    PPP connect address telephone user name (username) password
    (:password!)
    my timeCount(5)
    tell application "NetEvents 1.0b6"
      activate
      set NEW_AD_TXT to AddrToName (MyIPAddress)
      quit
    end tell
  on error errmsg number errnum
    if (state of (PPP status)) is "connected" then
      try
        tell application "NetEvents 1.0b6" to quit
        PPP disconnect
      on error
        end try
    end if
    my timeCount(5)
  end try
end repeat
```

```
-----htmlの書き換え
tell application "スクリプティング対応エディタ"
  open alias "VegaHD:/x/sunrise:/index.html"
  tell document 1
    tell paragraph 77
      cut
      make paragraph with data "red_arrow.gif"
    end tell
    tell paragraph 80
      cut
      make paragraph with data NEW_AD_TXT
    end tell
    tell paragraph 83
      cut
      make paragraph with data "現在アクセス可能¥, 翌朝07:50まで!"
    end tell
    save in file "VegaHD:/x/sunrise:/index.html" as text
    close saving no
  end tell
  quit
end tell
```

```
-----htmlのアップロード
set gDesturl to "ftp://99999999:*****@www.age.ne.jp/" --(6)
set doc to alias "VegaHD:/x/sunrise:/index.html"
with timeout of 3000 seconds
  set SS to "No"
  repeat until SS = "OK"
    try
      tell application "Fetch 3.0.1"
        activate
        put into url gDesturl item doc
        quit
      end tell
      set SS to "OK"
    on error
      set SS to "No"
    end try
  end repeat
end timeout
```

```
-----WWWサーバー起動
tell application "Quid Pro Quo" to activate
-----ブラウザ起動
```

```
tell application "インターネットエクスプローラ 3.0.1"
  Activate
  repeat
    OpenURL "http://" & NEW_AD_TXT toWindow 2 Flags 1
    my timeCount(600)
    if (endTime < date1) and (date1 < strTime) then exit repeat
  end repeat
  quit
end tell
-----WWWサーバー終了
tell application "Quid Pro Quo" to quit
-----htmlの書き換え
tell application "スクリプティング対応エディタ"
  open alias "VegaHD:/x/sunrise:/index.html"
  tell document 1
    tell paragraph 77
      cut
      make paragraph with data "black_arrow.gif"
    end tell
    tell paragraph 80
      cut
      make paragraph with data "home.age.or.jp/x/sunrise/sorry.html"
    end tell
    tell paragraph 83
      cut
      make paragraph with data "現在アクセス不可, 23:10まで待て!"
    end tell
    save in file "VegaHD:/x/sunrise:/index.html" as text
    close saving no
  end tell
  quit
end tell
```

```
-----htmlのアップロード
set gDesturl to "ftp://99999999:*****@www.age.ne.jp/"--(6)
set doc to alias "VegaHD:/x/sunrise:/index.html"
with timeout of 3000 seconds
  set SS to "No"
  repeat until SS = "OK"
    try
      tell application "Fetch 3.0.1"
        activate
        put into url gDesturl item doc
        quit
      end tell
      set SS to "OK"
    on error
      set SS to "No"
    end try
  end repeat
end timeout
```

```
-----回線切断
repeat while ((state of (PPP status)) ≠ "idle")
  try
    PPP disconnect
  on error
    end try
end repeat
-----時間チェックルーチン
on timeCount(s)
  set intervalSecond to s
  set date0 to current date
  repeat while ((current date) - date0) < intervalSecond
    set date1 to time of date ((current date) as text)
    if (endTime < date1) and (date1 < strTime) then exit repeat
  end repeat
end timeCount
```

htmlの書き換えの内容は、各自のホームページに合わせて下さい。(このリストの内容は筆者のindex.htmlの書き換えるものです。)